Option Compare Database
Option Explicit
 

Sub Test1()
' Will produce a warning if the user's setting, under Tools > Options
' on the Edit/Find tab includes confirming Action Queries
 

    DoCmd.RunSQL "UPDATE tblPeople SET tblPeople.City = ""Renton"""
 

End Sub
 

Sub Test2()
' Includes turning off warnings and then immediately turning warnings
' back on. Note: Warnings will not be turned on if the DoCmd.RunSQL
' statement fails for any reason, such as a table being renamed!
 

    DoCmd.SetWarnings False
   DoCmd.RunSQL "UPDATE tblEmployees SET tblEmployees.City = ""Seattle"""
    DoCmd.SetWarnings True
 

End Sub
 

 

Sub Test3()
' Same as above, but warnings will always be turned back on in the error
' handling code. This is an example of how error handling procedures can
' help prevent problems.
 

On Error GoTo ProcError
 

    DoCmd.SetWarnings False
    DoCmd.RunSQL "UPDATE tblPeople SET tblPeople.City = ""Tukwila"""
    DoCmd.SetWarnings True
 

' Additional code can be inserted here after re-enabling warnings
 

ExitProc:
  DoCmd.SetWarnings True
  Exit Sub
ProcError:
  MsgBox "Error: " & Err.Number & ". " & Err.Description, _
          vbInformation, "Error in Test3 Procedure..."
  Resume ExitProc
End Sub
 

Sub Test4()
' This method uses DAO. It requires a reference set to the Microsoft DAO 3.x Object Library

‘ (Access 97 --> 3.51, Access 2000+ --> 3.6) if you include the optional parameter dbFailOnError.
‘dbFailOnError Rolls back updates if an error occurs (Microsoft Jet workspaces only).
On Error GoTo ProcError
 

Dim db As DAO.Database
Dim strSQL As String
 

Set db = CurrentDb()
strSQL = "UPDATE tblPeople SET tblPeople.City = ""Lynnwood"""
 

  db.Execute strSQL, dbFailOnError
 

ExitProc:
  ' Cleanup
  On Error Resume Next
  db.Close

  Set db = Nothing
  Exit Sub
ProcError:
  MsgBox "Error: " & Err.Number & ". " & Err.Description, _
          vbInformation, "Error in Test4 Procedure..."
  Resume ExitProc
End Sub 

 

Sub Test5()
' This method uses ADO. It requires a reference set to the Microsoft ActiveX
' Data Objects 2.x Library (2.1 is the default library for Access 2000+).
 

On Error GoTo ProcError
 

Dim strSQL As String
Dim lngRecordsAffected As Long
 

strSQL = "UPDATE tblPeople SET tblPeople.City = ""Bellevue"""
 

  CurrentProject.Connection.Execute strSQL, lngRecordsAffected, adCmdText


  MsgBox lngRecordsAffected & " records were successfully updated.", vbInformation
  
ExitProc:
  Exit Sub
ProcError:
  MsgBox "Error: " & Err.Number & ". " & Err.Description, _
          vbInformation, "Error in Test5 Procedure..."
  Resume ExitProc
End Sub
